A POLY-TIME KNOT POLYNOMIAL VIA SOLVABLE APPROXIMATION

DROR BAR-NATAN AND ROLAND VAN DER VEEN

ABSTRACT. We construct the first poly-time-computable knot polynomial since Alexander's [Al, 1928] by using some new commutator-calculus techniques and a Lie algebra $\mathfrak{g}_1$ which is at the same time solvable and an approximation of the simple Lie algebra sl_2.

CONTENTS

1. INTRODUCTION

An excellent classical paper in mathematics would state a great theorem in the first couple pages of the introduction, describe its significance in section 2, and then present its proof in the remaining sections. We cannot match that; as a "theorem", our result is that a certain algorithm works well. We cannot even state the theorem without describing the algorithm, and a wordy description of the algorithm would either be lengthy and unmotivated, or would take up a whole paper. So instead of a wordy description we provide a complete implementation, right below in Figure 1.1, along with one usage example. Once it is established that the algorithm fits in less than a page of code, we move on to explain why we think it is valuable, and more importantly, how it arises from a traditional construction applied to a non-traditional Lie algebra which enables a non-traditional computational technique.

Theorem 1.1. *The program in Figure 1.1 computes z_1, a polynomial invariant of knots, in time polynomial in the crossing number.*

computations below

Date: First edition Not Yet, 2017, this edition Jun. 24, 2017. Electronic version and related files at [BNVDV], http://drorbn.net/AcademicPensieve/Projects/PPSA/PPSA-170103.
2010 *Mathematics Subject Classification.* 57M25.
Key words and phrases. knots, tangles, knot polynomials, Lie algebras, Lie bialgebras.
This work was partially supported by NSERC grant RGPIN 262178.

A Demo Program for z_1

```
DP[x_→D[α_], y_→D[β_]][P_][f_] := (* means P[∂α,∂β][f] *)
  Total[CoefficientRules[P, {x, y}] /.
    ({m_, n_} → c_) :→ c D[f, {α, m}, {β, n}]];
CF[𝔼[ω_, L_, Q_, P_]] := Expand /@ Together /@
    𝔼[ω /. b :→ Log[t], L, Q /. b :→ Log[t],
     P /. b :→ Log[t]];
𝔼 /: 𝔼[ω1_, L1_, Q1_, P1_] 𝔼[ω2_, L2_, Q2_, P2_] :=
  CF@𝔼[ω1 ω2, L1 + L2, ω2 Q1 + ω1 Q2, ω2^4 P1 + ω1^4 P2];
N[u[i_] c[j_] → k_][𝔼[ω_, L_, Q_, P_]] :=
  With[{q = ℯ^-γ β u[k] + γ c[k]}, CF[
    𝔼[ω, γ c[k] + (L /. c[j] → 0), ω ℯ^-γ β u[k] + (Q /. u[i] → 0),
      ℯ^-q DP[c[j]→D[γ], u[i]→D[β]][P][ℯ^q]] /.
     {γ → ∂_{c[j]} L, β → ω^-1 ∂_{u[i]} Q}]];
N[w[i_] c[j_] → k_][𝔼[ω_, L_, Q_, P_]] :=
  With[{q = ℯ^γ α w[k] + γ c[k]}, CF[
    𝔼[ω, γ c[k] + (L /. c[j] → 0), ω ℯ^γ α w[k] + (Q /. w[i] → 0),
      ℯ^-q DP[c[j]→D[γ], w[i]→D[α]][P][ℯ^q]] /.
     {γ → ∂_{c[j]} L, α → ω^-1 ∂_{w[i]} Q}]];

N[w[i_] u[j_] → k_][𝔼[ω_, L_, Q_, P_]] :=
  With[{q = (1 - t) μ^-1 α β + μ^-1 β u[k] + μ^-1 δ u[k] w[k] + μ^-1 α w[k]},
   CF[
    𝔼[μ ω, L, μ ω q + μ (Q /. w[i] | u[j] → 0),
       μ^4 ℯ^-q DP[w[i]→D[α], u[j]→D[β]][P][ℯ^q] + ω^4 Λ[k]] /.
      μ → 1 + (t - 1) δ /.
     {α → ω^-1 (∂_{w[i]} Q /. u[j] → 0), β → ω^-1 (∂_{u[j]} Q /. w[i] → 0),
      δ → ω^-1 ∂_{w[i],u[j]} Q}]];
m[i_, j_ → k_][Z_] := Module[{x, y, z},
  Z // N[w[i] c[j] → x] // N[w[x] u[j] → y] //
     ReplaceAll[{c[x|y] → c[x], w[j] → w[y]}] // N[u[i] c[x] → x] //
    ReplaceAll[z_[i|j|x|y] → z[k]] // CF]
```

FIGURE 1.1. A demo program computing z_1, some initial data, and a sample run on the 0-framed trefoil knot. The program is written in *Mathematica* [Wo] and is available at [BNVDV].

Initial Data

```
Λ[k_] := (1 - t) (α^2 β^2 + 4 α β δ μ + 2 δ^2 μ^2) / 2 +
    2 μ^2 (α β + δ μ) c[k] - β (2 μ - 1) (α β + 2 δ μ) u[k] +
    2 β δ μ^2 c[k] u[k] - β^2 δ (3 μ - 1) u[k]^2 / 2 + α (α β + 2 δ μ) w[k] +
    2 α δ μ^2 c[k] w[k] - 2 (t - 1) δ^2 (α β + δ μ) u[k] w[k] +
    2 δ^2 μ^2 c[k] u[k] w[k] - β δ^2 (2 μ - 1) u[k]^2 w[k] + α^2 δ (1 + μ) w[k]^2 / 2 +
    α δ^2 u[k] w[k]^2 - (t - 1) δ^4 u[k]^2 w[k]^2 / 2;
R⁺[i_, j_] := 𝔼[1, b[i] c[j], u[i] w[j],
    -c[i] (t - 1)^2 / 2 - c[i]^2 (t - 1)^2 / 2 + c[i] c[j] (t^2 - t - 2) / 2 -
     c[j] u[i] w[i] / 2 + c[i] (1 - t) u[i] w[i] - u[i]^2 w[i]^2 / 2 + u[i] w[j] +
     c[j] t u[i] w[j] / 2 + c[i] (t - 2) t u[i] w[j] + c[i] (1 + t) u[j] w[j] / 2 +
     (t - 1) u[i]^2 w[i] w[j] - (t - 2) t u[i]^2 w[j]^2 / 2];
R⁻[i_, j_] := 𝔼[1, -b[i] c[j], -t^-1 u[i] w[j],
    c[i] (t - 1)^2 / 2 + c[i]^2 (t - 1)^2 / 2 + c[i] c[j] (2 + t - t^2) / 2 +
     c[j] u[i] w[i] / 2 + c[i] (t - 1) u[i] w[i] + u[i]^2 w[i]^2 / 2 +
     (1 - t^-1) u[i] w[j] / 2 + c[i] (2 t - 5 + 3 t^-1) u[i] w[j] / 2 +
     c[j] (t^-1 + 1 - t) u[i] w[j] / 2 - c[i] (t + 1) u[j] w[j] / 2 +
     (2 - 3 t^-1) u[i]^2 w[i] w[j] / 2 + (1 + 2 t^-2 - 3 t^-1) u[i]^2 w[j]^2 / 2 -
     t^-1 (1 + t) u[i] u[j] w[j]^2 / 2];
ur[i_] := 𝔼[t^(-1/4), 0, 0, c[i] t / 4 + u[i] w[i] / 8];
nr[i_] := 𝔼[t^(1/4), 0, 0, -c[i] t^3 / 4 - t^2 u[i] w[i] / 8];
ul[i_] := 𝔼[t^(1/4), 0, 0, c[i] t (4 + t) / 4 - t^2 u[i] w[i] / 8];
nl[i_] := 𝔼[t^(-1/4), 0, 0, -c[i] (1 + 4 t^-1) / 4 + u[i] w[i] / 8];
```

The Trefoil

```
z = R⁺[1,14] R⁻[5,2] nr[3] ul[4] R⁺[19,6] R⁻[7,10] nl[8] ur[9] R⁺[11,20] nr[12] ul[13]
    R⁻[15,18] nl[16] ur[17];
(Do[z = z // m[1,k→1], {k, 2, 20}]; z = z /. a_1 :→ a)
```

$$\mathbb{E}\Big[-1+\tfrac{1}{t}+t,\ 0,\ 0,\ -16+\tfrac{9c}{2}-\tfrac{2c}{t^4}+\tfrac{1}{t^3}+\tfrac{11c}{2t^3}-\tfrac{4}{t^2}-\tfrac{8c}{t^2}+\tfrac{10}{t}+\tfrac{4c}{t}+18t-10ct-14t^2+8ct^2+7t^3-\tfrac{3ct^3}{2}-2t^4-2ct^4+2ct^5-\tfrac{ct^6}{2}-4uw+\tfrac{2uw}{t^4}-\tfrac{7uw}{2t^3}+\tfrac{9uw}{2t^2}+\tfrac{uw}{2t}+6tuw-2t^2uw-\tfrac{1}{2}t^3uw+\tfrac{3}{2}t^4uw-\tfrac{1}{2}t^5uw\Big]$$

The invariant z_1 is in fact quite strong — explicit computations show that it separates more knots in the standard knot tables than the Alexander polynomial, the Jones polynomial, the HOMFLY-PT polynomial, and Khovanov homology (even when these are combined)! See the table in Section 5.

This is exciting news. The main reason is self-evident — z_1 is the first poly-time computable polynomial invariants of knots since the Alexander polynomial, discovered nearly 90 years ago [Al, 1928]. A further reason is explained in Section 6 — it seems that z_1 has a better chance than anything else we know to detect potential counterexamples to the ribbon-slice conjecture. Finally, z_1 is founded on some new Lie-theoretic techniques, it seems to be possible to generalize it in several directions, and many questions are raised. See Section 7.

Our z_1 can be considered as a perturbation of another invariant, z_0, which is somewhat weaker yet in many ways it is more valuable. It is computed by a shorter program (see Figure 3.1) which runs faster, and it has a proven topological significance. The main reason z_0 is not the main character of this paper is that it is already well known: restricted to knots z_0 is the Alexander polynomial and considered on pure tangles (see Section 3) it is equivalent to the "β-calculus" of [BNS, BN2], which in itself is mostly equivalent [Ha] to Archibald's calculus [Ar] and is a mild extension of the Burau-Gassner theory of [LD, KLW, CT, CC].

As we shall see in Section 4, z_1 arises from a certain Lie algebra $\mathfrak{g}_1$, and as we shall see in Section 3, z_0 arises from a certain Lie algebra $\mathfrak{g}_0$. Within this introduction we only wish to define these two Lie algebras and explain how they arise as "(solvable) approximations of sl_2".

Definition 1.2. Over the field $\mathbb{Q}$ of rational numbers, let $\mathfrak{g}_0$ be the 4-dimensional Lie algebra generated by elements b, c, u, and w, such that b is central, $[c,u]=u$, $[c,w]=-w$, and $[u,w]=b$:

$$\mathfrak{g}_0 := \mathbb{Q}\langle b,c,u,w\rangle/([b,-]=0,\ [c,u]=u,\ [c,w]=-w,\ [u,w]=b)\ .$$

We grade $\mathfrak{g}_0$ by declaring that $\deg(b,c,u,w)=(1,0,1,0)$.

Note that $\mathfrak{g}_0$ is solvable: Indeed

$$\mathfrak{g}_0 \supset \langle b,u,w\rangle \supset \langle b\rangle \supset \{0\}$$

is a tower of ideals of $\mathfrak{g}_0$ whose consequetive quotients are Abelian.

Definition 1.3. Over the ring $R := \mathbb{Q}[\epsilon]/(\epsilon^2=0)$, let $\mathfrak{g}_1$ be the 4-dimensional Lie algebra generated by elements b, c, u, and w, such that b is central, $[c,u]=u$, $[c,w]=-w$, and $[u,w]=b-2\epsilon c$:

$$\mathfrak{g}_1 := R\langle b,c,u,w\rangle/([b,-]=0,\ [c,u]=u,\ [c,w]=-w,\ [u,w]=b-2\epsilon c)\ .$$

We grade R and $\mathfrak{g}_1$ by declaring that $\deg(b,c,u,w,\epsilon)=(1,0,1,0,1)$.

Note that over $\mathbb{Q}$, $\mathfrak{g}_1$ is an 8-dimensional solvable Lie algebra: Indeed

$$\mathfrak{g}_1 \supset \langle b,u,w,\epsilon b,\epsilon c,\epsilon u,\epsilon w\rangle \supset \langle b,\epsilon b,\epsilon c,\epsilon u,\epsilon w\rangle \supset 0$$

is a tower of ideals of $\mathfrak{g}_1/\mathbb{Q}$ whose consequetive quotients are Abelian.

For the purposes of this paper, we do not need to know that $\mathfrak{g}_0$ and $\mathfrak{g}_1$ are related to the classical Lie algebra sl_2, and hence the next few paragraphs[1] can safely be skipped. Yet these paragraphs say that "there's more where $\mathfrak{g}_0$ and $\mathfrak{g}_1$ came from", and strongly suggest that there's more where our z_0 and z_1 come from.

Let $\mathfrak{g}$ be an arbitrary semi-simple Lie algebra over $\mathbb{Q}$, and let $\mathfrak{g}=\mathfrak{n}^+\oplus\mathfrak{h}\oplus\mathfrak{n}^-$ be MORE.

2. Algebras, Yang-Baxter Elements and Spinners, Invariants

MORE.

[1]Trigger warning: Borel and Cartan subalgebras, Lie bialgebras, Drinfel'd doubles.

A Demo Program for z_0

```
CF[ω_. 𝔼[Q_]] := Simplify[ω] 𝔼[Simplify[Q]];
𝔼 /: 𝔼[Q1_] 𝔼[Q2_] := CF@𝔼[Q1 + Q2];
N_{u_i_ c_j_→k_}[ω_. 𝔼[Q_]] :=
  CF[ω 𝔼[ⅇ^{-γ} β u_k + γ c_k + (Q /. c_j | u_i → 0)] /.
    {γ → ∂_{c_j} Q, β → ∂_{u_i} Q}];
N_{w_i_ c_j_→k_}[ω_. 𝔼[Q_]] :=
  CF[ω 𝔼[ⅇ^γ α w_k + γ c_k + (Q /. c_j | w_i → 0)] /.
    {γ → ∂_{c_j} Q, α → ∂_{w_i} Q}];
N_{w_i_ u_j_→k_}[ω_. 𝔼[Q_]] :=
  CF[
   ν ω 𝔼[-b ν α β + ν β u_k + ν δ u_k w_k + ν α w_k +
       (Q /. w_i | u_j → 0)] /. ν → (1 + b δ)^{-1} /.
    {α → ∂_{w_i} Q /. u_j → 0, β → ∂_{u_j} Q /. w_i → 0, δ → ∂_{w_i,u_j} Q}];
m_{i_,j_→k_}[ω_. 𝔼[Q_]] :=
 CF[Module[{x},
   (ω 𝔼[Q] // N_{w_i c_j→x} // N_{u_i c_x→x} // N_{w_x u_j→x}) /.
    {c_i → c_k, w_j → w_k, y_x_ :→ y_k}]]
```

Initial Data

```
R^+_{i_,j_} := 𝔼[b c_j + b^{-1} (ⅇ^b - 1) u_i w_j];
R^-_{i_,j_} := 𝔼[-b c_j + b^{-1} (ⅇ^{-b} - 1) u_i w_j];
```

The Knot 8_{17}

```
z = R^-_{12,1} R^-_{2,7} R^-_{8,3} R^-_{4,11} R^+_{16,5} R^+_{6,13} R^+_{14,9} R^+_{10,15};
Do[z = z // m_{1,n→1}, {n, 2, 16}];
CF@z /. ω_. 𝔼[Q_] :→ ω^{-1} /. b → Log[t]
```

$$-\frac{1-4t+8t^2-11t^3+8t^4-4t^5+t^6}{t^3}$$

Figure 3.1. A demo program computing z_0, some initial data, and a sample run on the knot 8_{17}. The program is written in *Mathematica* [Wo] and is available at [BNVDV].

3. The Lie Algebra $\mathfrak{g}_0$, the Invariant z_0, and the Alexander Polynomial

4. The Lie Algebra $\mathfrak{g}_1$ and the Invariant z_1

MORE.

5. Computing z_1

MORE.

6. An Aside on Ribbon Knots

MORE.

7. Questions

MORE.

References

[Al] J. W. Alexander, *Topological invariants of knots and link,* Trans. Amer. Math. Soc. **30** (1928) 275–306. Page(s) 1, 2.

[Ar] J. Archibald, *The Multivariable Alexander Polynomial on Tangles,* Ph.D. thesis, University of Toronto, 2010. See `http://www.math.toronto.edu/~jfa/`. Page(s) 3.

[BN1] D. Bar-Natan, *On the Vassiliev knot invariants,* Topology **34** (1995) 423–472. Page(s) .

[BN2] D. Bar-Natan, *Balloons and Hoops and their Universal Finite Type Invariant, BF Theory, and an Ultimate Alexander Invariant,* Acta Mathematica Vietnamica **40-2** (2015) 271–329, arXiv:1308.1721. Page(s) 3.

[BNS] D. Bar-Natan and S. Selmani, *Meta-Monoids, Meta-Bicrossed Products, and the Alexander Polynomial,* J. of Knot Theory and its Ramifications **22-10** (2013), arXiv:1302.5689. Page(s) 3.

[BNVDV] D. Bar-Natan and Roland van der Veen, *A Poly-Time Knot Polynomial Via Solvable Approximation,* (self-reference), paper and related files at http://drorbn.net/AcademicPensieve/Projects/PPSA. The arXiv:????.????? edition may be older. Page(s) 1, 2, 4.

[CC] D. Cimasoni and A. Conway, *A Burau-Alexander 2-functor on Tangles,* arXiv:1606.05166. Page(s) 3.

[CT] D. Cimasoni and V. Turaev, *A Lagrangian Representation of Tangles,* Topology **44** (2005) 747–767, arXiv:math.GT/0406269. Page(s) 3.

[EK] P. Etingof and D. Kazhdan, *Quantization of Lie Bialgebras, I,* Selecta Mathematica, New Series **2** (1996) 1–41, arXiv:q-alg/9506005. Page(s) .

[Ha] I. Halacheva, *Alexander Type Invariants of Tangles, Skew Howe Duality for Crystals and The Cactus Group,* Ph.D. thesis, University of Toronto, 2016. Page(s) 3.

[Ka] L. H. Kauffman, *Rotational Virtual Knots and Quantum Link Invariants,* arXiv:1509.00578. Page(s) .

[KLW] P. Kirk, C. Livingston, and Z. Wang, *The Gassner Representation for String Links,* Comm. Cont. Math. **3** (2001) 87–136, arXiv:math/9806035. Page(s) 3.

[LD] J. Y. Le Dimet, *Enlacements d'Intervalles et Représentation de Gassner,* Comment. Math. Helv. **67** (1992) 306–315. Page(s) 3.

[Wo] *Wolfram Language & System Documentation Center,* http://reference.wolfram.com/language. Page(s) 2, 4.

Department of Mathematics, University of Toronto, Toronto Ontario M5S 2E4, Canada
E-mail address: drorbn@math.toronto.edu
URL: http://www.math.toronto.edu/~drorbn

Mathematisch Instituut, Universiteit Leiden, Niels Bohrweg 1, 2333 CA Leiden, The Netherlands
E-mail address: roland.mathematics@gmail.com
URL: http://www.rolandvdv.nl/